

## Tableurs - Principales fonctions traitement de chaînes de caractères

Voici une liste non exhaustive des principales fonctions utilisables pour le traitement de chaînes de caractères dans le cadre des besoins liés à la généalogie.

Vous avez par exemple récupéré une longue liste de patronymes et prénoms, ou toute autre liste, et vous désirez y apporter des mises en forme que ne vous voulez pas faire manuellement. La mise en œuvre de fonctions de traitement de chaînes de caractère va vous permettre de le faire automatiquement.

### Rappels

Un calcul effectué dans une cellule peut utiliser une ou plusieurs fonctions.

Lors de son entrée dans une cellule, le calcul doit être précédés d'un "=". Ceci pour la première fonction seulement si le calcul effectué en utilise plusieurs.

Les paramètres de chacune des différentes fonctions sont placés entre les parenthèses et séparés par un point virgule.

- Cas d'un calcul utilisant une seule fonction : `=FONCTION(paramètre1;paramètre2;paramètre3)`
- Cas d'un calcul utilisant plusieurs fonctions :  
`=FONCTION1(paramètre1_de_la_fonction1;FONCTION2(paramètre1_de_la_fonction_2;paramètre_2_de_la_fonction2);paramètre3_de_la_fonction1)`

Dans ce cas, la partie soulignée représente le paramètre2 de la fonction1 qui est réalisé par la fonction2 et ses propres paramètres.



Le nombre de parenthèses ouvrantes doit être égal au nombre des parenthèses fermantes.

Les paramètres peuvent être entrés de différentes manières :

- Un/des paramètres peuvent être eux-mêmes une/des fonctions.
- Du texte encadré par deux double quotes. Ex. : "maison"
- Une valeur numérique.
- Issus du contenu d'une cellule ou une plage de cellules pointées par ses coordonnées numéro-colonne/numéro-ligne. Ex. : **B4** ou **\$C25**, ...
- Issus du contenu d'une cellule ou d'une plage de cellules pointées par le *nom* qui leur aura été précédemment attribué. Ex. : *lieu\_naissance*, *prenom*, ...
- Le panachage de ces possibilités est possible.

Ceci pouvant varier suivant les fonctions.



Dans les tableurs, lors de la mise en œuvre d'une fonction, un tableau de saisie vous aide : chaque paramètre à documenter est expliqué et détaillé, les pré-résultats obtenus sont donnés au fur et à mesure.

### Fonctions de traitement de chaînes de caractères

Reportez-vous aux rappels en haut du document pour le contenu des paramètres.

#### CONCATENER

La fonction CONCATENER permet d'assembler des chaînes de caractères.

`=CONCATENER(chaîne1;chaîne2;chaîne3;chaîne4)`

#### &

Pour la concaténation, on peut également utiliser l'opérateur &

`="chaîne1"&"chaîne2"&"chaîne3"&"chaîne4"`

Ex. : `="Bordeaux, "&"Gironde"`  $\Rightarrow$  On obtiendra : "Bordeaux, Gironde"

#### DROITE

La fonction DROITE permet d'extraire un nombre de caractères désiré à partir de la droite (fin) d'une "chaîne traitée".

`=DROITE(chaîne_traitée;nbre_caractères)`

Ex. : `=DROITE("Bordeaux, Gironde";4)`  $\Rightarrow$  On obtiendra : "onde"

## GAUCHE

La fonction GAUCHE permet d'extraire un nombre de caractères désiré à partir de la gauche d'une "chaîne\_traitée".

**=GAUCHE(chaîne\_traitée;nbre\_caractères)**

Ex. : =GAUCHE("Bordeaux, Gironde";4) ➡ On obtiendra : "Bord"

## STXT

La fonction STXT permet d'extraire un nombre de caractères désiré vers la droite à partir d'un caractère précisé par sa position, d'une "chaîne\_traitée".

**=STXT(chaîne\_traitée;position\_départ;nbre\_caractères)**

Ex. : =STXT("Bordeaux, Gironde";5;3) ➡ On obtiendra : "eau"

## NBCAR

La fonction NBCAR permet de connaître le nombre de caractères contenu dans une "chaîne\_traitée".

**=NBCAR(chaîne\_traitée)**

Seront comptés avec les caractères alphanumériques, tous les éléments tels que les espaces et signes de ponctuation...

Ex. : =NBCAR("Bordeaux, Gironde") ➡ On obtiendra : 17

## TROUVE

La fonction TROUVE permet de connaître la position depuis la gauche du premier élément d'une "chaîne\_cherchée" dans une "chaîne\_traitée" pouvant la contenir. Cette fonction est sensible à la casse contrairement à la fonction CHERCHE qui ne l'est pas.

**=TROUVE(chaîne\_cherchée;chaîne\_traitée;position\_départ)**

"position\_départ" est la position à partir de laquelle on effectuera la recherche. C'est un paramètre optionnel, par défaut = 1 si non renseigné. Ce paramètre peut servir à chercher la position d'une deuxième chaîne\_cherchée en démarrant la recherche depuis la position de la précédente, si elle existe.

Ex. : =TROUVE("eau";"Bordeaux, Gironde") ➡ On obtiendra : 5



Si la chaîne\_cherchée n'est pas présente dans la chaîne\_traitée, alors renvoi de #VALEUR! qui est à traiter à l'aide de SIERREUR ou ESTERREUR (voir paragraphe **Traitement de l'erreur #VALEUR!**).

## CHERCHE

La fonction CHERCHE permet de connaître la position depuis la gauche du premier élément d'une "chaîne\_cherchée" dans une "chaîne\_traitée" pouvant la contenir. Cette fonction est insensible à la casse contrairement à la fonction TROUVE qui l'est.

**=CHERCHE(chaîne\_cherchée;chaîne\_traitée;position\_départ)**

"position\_départ" est la position à partir de laquelle on effectuera la recherche. C'est un paramètre optionnel, par défaut = 1 si non renseigné.

Ex. : =CHERCHE("eau";"Bordeaux, Gironde") ➡ On obtiendra : 5



Si la chaîne\_cherchée n'est pas présente dans la chaîne\_traitée, alors renvoi de #VALEUR! qui est à traiter à l'aide de SIERREUR ou ESTERREUR (voir paragraphe **Traitement de l'erreur #VALEUR!**).

## SUPPRESPEACE

La fonction SUPPRESPEACE permet d'éliminer tous les espaces superflus d'une "chaîne\_traitée".

**=SUPPRESPEACE(chaîne\_traitée)**

Ex. : =SUPPRESPEACE("Bordeaux, Gironde ") ➡ On obtiendra : "Bordeaux, Gironde"



S'il la "chaîne\_traitée" en contient, la fonction laissera un espace avant et après les signes de ponctuation, sauf le final qui n'en aura pas après lui.

Ex. : =SUPPRESPEACE("Bordeaux, Gironde .") ➡ On obtiendra : "Bordeaux, Gironde."

1 espace avant la virgule      1 espace avant le point

## MAJUSCULE

La fonction MAJUSCULE convertit tout de contenu de la "chaîne\_traitée" en caractère majuscules.

**=MAJUSCULE(chaîne\_traitée)**

Ex. : =MAJUSCULE("Bordeaux, Gironde") ➡ On obtiendra : "BORDEAUX, GIRONDE"

## MINUSCULE

La fonction MINUSCULE convertit tout de contenu de la "chaîne\_traitée" en caractère minuscules.

**=MINUSCULE(chaîne\_traitée)**

Ex. : =MINUSCULE("Bordeaux, Gironde")  $\Rightarrow$  On obtiendra : "bordeaux, gironde"

## NOMPROPRE

La fonction NOMPROPRE convertit en majuscule la première lettre de tous les mots de la "chaîne\_traitée".

**=NOMPROPRE(chaîne\_traitée)**

Ex. : =NOMPROPRE("bordeaux, gironde")  $\Rightarrow$  On obtiendra : "Bordeaux, Gironde"

## SUBSTITUE

Dans une "chaîne\_traitée", la fonction SUBSTITUE permet de remplacer les occurrences d'une "chaîne\_à\_substituer" par une "chaîne\_de\_substitution".

**=SUBSTITUE(chaîne\_traitée;chaîne\_à\_substituer;chaîne\_de\_substitution;[numéro\_occurrence])**

"numéro\_occurrence" indique le numéro de l'occurrence de chaîne\_à\_substituer dans la chaîne\_traitée s'il y en a plusieurs, paramètre optionnel. Si ce paramètre est omis, la fonction remplacera chaque occurrence de la chaîne\_à\_substituer trouvée.

Ex. : =SUBSTITUE("J.P."; "."; ". ";1)  $\Rightarrow$  On obtiendra : "J. P.". *Nous avons rajouté un espace après le premier point entre J. et P.*

## REEMPLACER

La fonction REMPLACER permet de remplacer un "nbre\_caractères" désiré à partir d'une "position\_départ" à l'intérieur d'une "chaîne\_traitée", par une "chaîne\_de\_substitution".

**=REEMPLACER(chaîne\_traitée;position\_départ;nbre\_caractères;chaîne\_de\_substitution)**

La "chaîne\_traitée" peut être directement une chaîne de caractères (entre des double quotes), le contenu d'une cellule pointée par son adresse ou le nom qui lui aura été précédemment attribué.

"chaîne\_de\_substitution" est la chaîne qui va remplacer le "nbre\_caractères" désiré dans la "chaîne\_traitée". Leur nombre de caractères peut être différent.

Le "nbre\_caractères" et la "chaîne\_traitée" peuvent être une valeur numérique, le résultat d'un calcul réalisé avec une ou plusieurs fonctions, le contenu d'une cellule pointée par son adresse ou par son nom.

Ex. : =REEMPLACER("bateau";2;2;"oul")  $\Rightarrow$  On obtiendra : "bouleau"

## EXACT

La fonction EXACT vérifie si les contenus de 2 cellules sont rigoureusement identiques (sensible à la casse).

**=EXACT(chaîne\_traitée1;chaîne\_traitée2)**

La fonction renvoie VRAI ou FAUX.

Ex. : =EXACT("Bordeaux","bordeaux")  $\Rightarrow$  On obtiendra : FAUX

On utilise cette fonction dans un test conditionnel :

**=SI(EXACT(chaîne\_traitée1;chaîne\_traitée2);quoi\_faire\_si\_vrai;quoi\_faire\_si\_faux)**

Les "chaîne\_traitée1" "chaîne\_traitée2" peuvent être directement des chaînes de caractères (entre des double quotes), le contenu d'une cellule pointée par son adresse ou le nom qui lui aura été précédemment attribué.

## Fonctions logiques

### SI

La fonction SI permet de réaliser une action en fonction du résultat du test d'une condition.

**=SI(condition;quoi\_faire\_si\_condition\_VRAI;quoi\_faire\_si\_condition\_FAUX)**

La "condition" peut être un calcul réalisé avec une ou plusieurs fonctions, une valeur numérique, le contenu d'une cellule pointée par son adresse ou par son nom. Le calcul peut être lui-même la comparaison entre deux valeurs, la valeur dans une autre cellule, ...

"Quoi\_faire\_si\_VRAI" ou "Quoi\_faire\_si\_FAUX" peuvent être un calcul réalisé avec une ou plusieurs fonctions, une valeur alphanumérique, le contenu d'une cellule pointée par son adresse ou par son nom.

**OU**

La fonction OU permet de regrouper un ensemble d'éléments qui sera validée si au minimum l'un d'eux est VRAI.

=OU(élément1;élément2;élément3)

Les éléments peuvent être un calcul réalisé par une ou plusieurs fonctions ou le contenu d'une cellule pointée par son adresse ou par son nom.

**ET**

La fonction ET permet de regrouper un ensemble d'éléments qui sera validée si tous sont VRAI.

=ET(élément1;élément2;élément3)

Chaque élément peut être un calcul réalisé par une ou plusieurs fonctions ou le contenu d'une cellule pointée par son adresse ou par son nom.

**NON**

La fonction NON permet d'inverser un résultat.

=NON(élément)

L'élément peut être un calcul réalisé par une ou plusieurs fonctions ou le contenu d'une cellule pointée par son adresse ou par son nom. Ce qui est VRAI devient FAUX et inversement.

**Traitement de l'erreur #VALEUR!****SIERREUR**

La fonction SIERREUR permet d'assigner une valeur dans le cas où le calcul exécuté dans une cellule retourne une erreur #VALEUR!.

=SIERREUR(calcul;valeur\_si\_erreur\_calcul)

"valeur\_si\_erreur\_calcul" peut être une valeur alpha numérique, l'adresse ou le nom d'une cellule.

Exemple de traitement : On veut trouver la position d'un caractère dans une chaîne à traiter à l'aide de la fonction TROUVE, et s'il n'y est pas présent on va, par exemple, choisir de retourner "0" si la fonction génère une erreur.

=SIERREUR(TROUVE("x";"Bordeaux, Gironde";1);0) ⇒ On obtiendra 8 si on cherche "x"

=SIERREUR(TROUVE("z";"Bordeaux, Gironde";1);0) ⇒ On obtiendra 0 si on cherche "z"

**ESTERREUR**

La fonction ESTERREUR retourne VRAI s'il y a une erreur dans un calcul, sinon retourne FAUX.

=ESTERREUR(calcul)

Associée à la fonction SI, la fonction ESTERREUR permet d'assigner une valeur dans le cas où le calcul exécuté dans une cellule retourne une erreur #VALEUR!.

On pourra traiter l'erreur par : =SI(ESTERREUR(calcul);valeur\_si\_VRAI;calcul)

**Calculs associant plusieurs fonctions****Compter le nombre d'occurrences d'un caractère dans une chaîne**

Pour trouver le nombre de fois où une occurrence "*caractère\_cherché*" apparaît dans une "*chaîne\_traitée*", on soustrait du nombre de caractères contenus dans la "*chaîne\_traitée*", le nombre de caractères de cette même chaîne où on a substitué le "*caractère\_cherché*" par "". (" = rien).

=NBCAR(chaîne\_traitée)-NBCAR(SUBSTITUE(chaîne\_traitée;caractère\_cherché;""))

Ex. : =NBCAR("Bordeaux, Gironde")-NBCAR(SUBSTITUE("Bordeaux, Gironde";"o";"")) ⇒ On obtient = 2



**Attention** si vous recherchez une chaîne et non un caractère !

**Compter le nombre de mots dans une chaîne**

Pour trouver le nombre de mots dans une "*chaîne\_traitée*", on soustrait du nombre de caractères contenus dans la "*chaîne\_traitée*", le nombre d'espaces contenu +1. On obtient le nombre d'espaces en comptant les substitutions de chaque " " (espace) par "" (rien).

=NBCAR(chaîne\_traitée)-NBCAR(SUBSTITUE(chaîne\_traitée;" ";""))+1

Ex. : =NBCAR("la petite maison dans la prairie")-NBCAR(SUBSTITUE("la petite maison dans la prairie";" ";""))+1

⇒ On obtient = 6

## Séparer une chaîne de caractères et en récupérer la partie gauche

Soit à séparer une chaîne de caractères en 2 parties, pour en récupérer la partie à gauche d'un élément "séparateur" choisi. S'il a plusieurs occurrences du "séparateur", il ne prendra en compte que la position de la première. Voir **Note** ci-dessous.

Dans le calcul proposé, l'élément séparateur n'est pas récupéré dans l'extraction.

**=GAUCHE(chaîne\_traitée;TROUVE(séparateur;chaîne\_traitée)-1))**

"séparateur" est l'élément à partir duquel la séparation va s'effectuer. Il peut comporter un ou plusieurs caractères. L'ajout de "-1" permet d'arrêter l'extraction effectuée par GAUCHE juste avant la position du premier élément du premier "séparateur" donnée par TROUVE, quelque soit le nombre de caractères de ce dernier.

Ex. : =GAUCHE("Bordeaux, Gironde";TROUVE(", ";"Bordeaux, Gironde")-1))

Ici le "séparateur" cherché est "virgule espace" (il contient 2 caractères).  $\Rightarrow$  On obtient : "Bordeaux"

## Séparer une chaîne de caractères et en récupérer la partie droite

Soit à séparer une chaîne de caractères en 2 parties, pour en récupérer la partie à droite d'un élément "séparateur" choisi. S'il a plusieurs occurrences du "séparateur", il ne prendra en compte que la position de la première. Voir **Note** ci-dessous.

Dans le calcul proposé, l'élément séparateur n'est pas récupéré dans l'extraction.

**=DROITE(chaîne\_traitée;NBCAR(chaîne\_traitée)-TROUVE(séparateur;chaîne\_traitée)-(NBCAR(séparateur)-1))**

"séparateur" est l'élément après lequel la séparation va s'effectuer. Il peut comporter un ou plusieurs caractères.

Le calcul "**-(NBCAR(séparateur)-1)**" permet de démarrer l'extraction effectuée par DROITE juste après la position du dernier caractère du premier "séparateur" donnée par TROUVE, quelque soit le nombre de caractères de ce dernier. Il peut être omis si le "séparateur" ne comprend qu'un seul caractère.

Ex. : =DROITE("Bordeaux, Gironde";NBCAR("Bordeaux, Gironde")-TROUVE(", ";"Bordeaux, Gironde")-(NBCAR(", ") -1))

Ici le "séparateur" cherché est "virgule espace" (il contient 2 caractères).  $\Rightarrow$  On obtient : "Gironde"

**Note** - On peut traiter le cas où il existe plusieurs occurrences de l'élément séparateur, la fonction TROUVE le permet avec le paramètre "position\_départ" qui n'est pas présent dans les deux cas exposés ci-dessus, il est optionnel. Voir la description de la fonction **TROUVE** en page 2 de ce document.

## Vérifier qu'une chaîne de caractères commence par une majuscule, est toute en majuscules ou toute en minuscules

Nous pourrions vérifier si le contenu d'une cellule commence pas une majuscule par la fonction **NOMPROPRE**, est tout en minuscule par la fonction **MINUSCULE** ou tout en majuscule par la fonction **MAJUSCULE**, chacune combinée avec la fonction **EXACT**.

**=SI(EXACT(chaîne\_traitée;NOMPROPRE(chaîne\_traitée));quoi\_faire\_si\_vrai;quoi\_faire\_si\_faux)**

**=SI(EXACT(chaîne\_traitée;MAJUSCULE(chaîne\_traitée));quoi\_faire\_si\_vrai;quoi\_faire\_si\_faux)**


**=SI(EXACT(chaîne\_traitée;MINUSCULE(chaîne\_traitée));quoi\_faire\_si\_vrai;quoi\_faire\_si\_faux)**

Ex. : Vérification que "bordeaux" commence par un "B":

=SI(EXACT("bordeaux";NOMPROPRE("bordeaux"));"Bien écrit";"mal écrit")  $\Rightarrow$  On obtient : "mal écrit"


## Structure d'un empilement de fonctions/calculs conditionnels

Pour effectuer un calcul complexe dans une cellule, nous pouvons réaliser un empilement de fonctions et/ou de calculs. Cas d'un empilement de 3 niveaux de tests conditionnels utilisant la fonction SI.



Ce qui donne :

```
=SI(cond1;action_si_cond1_vraie;SI(cond2;action_si_cond2_vraie;SI(cond3;action_si_cond3_vraie;action_si_cond3_fausse)))
```



A étendre autant que de besoin, mais attention à rester maître de la situation !

## Quelques conseils

- Il n'est pas conseillé de chercher à réaliser un calcul incluant plusieurs fonctions d'un seul coup, sauf à être sûr de vous.
- Quand vous entrez une fonction quelque soit son emplacement (après un =, ou en lieu et place d'un paramètre d'une autre fonction), après avoir entré son nom, placez aussitôt les deux parenthèses, puis tous les points virgule sans aucune valeur dans les parenthèses.  
Tapez alors sur entrée. Le tableur ne "dira rien" même si les paramètres sont vides avec les parenthèses et points virgule correctement en place, mais "râlera" si l'un deux est manquant ou superflu !  
Entrez alors les paramètres de chaque fonction.
- Avancez étape par étape en réalisant des calculs intermédiaires progressifs dans des cellules séparées qui vous mèneront au résultat final et vous permettront de vérifier les résultats au fur et à mesure de votre réalisation. Dans un deuxième temps introduisez les calculs ou fonctions intermédiaires réalisées dans le dernier calcul.
- **Attribuez des noms de cellule/plages de cellules en relation avec leur contenu. Ceci vous aidera à réaliser la mise en œuvre du calcul, vous vous y retrouverez beaucoup plus facilement.**
- Sauf à le connaître par cœur, vérifiez toujours l'ordre des paramètres d'une fonction, cela peut varier d'une fonction à une autre (inversion de position).
- Dans un calcul ou une fonction, le nombre de parenthèses ouvrantes doit être égal au nombre de parenthèses fermantes.



Certaines fonctions peuvent être présentes ou pas suivant votre tableur et sa version.

*Le contenu de ce document a été limité à quelques besoins associés à la généalogie dans le cadre des exposés réalisés par l'Auteur. Ceci n'est donc pas un document complet de référence pour l'usage du tableur Calc de LibreOffice, d'Excel ou d'un autre tableur.*

Pour toute question : [www.rakforgeron.fr/contact](http://www.rakforgeron.fr/contact)